Límites, diferenciación e integración

Definición dunha función. Diferencia cunha expresión

En MAPLE as relacións funcionais pódense definir de dous xeitos:
1. Mediante unha expresión ou fórmula
2. Cunha función matemática propiamente dita
Se temos unha expresión (caso 1):

```
> f:=x^3+1;
```

$$f := x^3 + 1 \tag{1.1.1}$$

e a queremos evaluar nun punto x=3, necesítase utiliza-lo comando **subs**

```
> subs(x=3, f); # substitúe x=3 na expresión f
```

$$28 \tag{1.1.2}$$

Para definir unha función propiamente dita (caso 2) utilizase o **operador frecha (->)**

```
> f:=x->x^3+1;f(3); #evaluamos f en x=3
```

$$f := x \rightarrow x^3 + 1$$

$$28 \tag{1.1.3}$$

A función tamén pode ter varias variábeis:

> $f := (x, y) \rightarrow x + y$

$$f := (x, y) \mapsto x + y \tag{1.1.4}$$

> $f(x, y)$

$$x + y \tag{1.1.5}$$

Tamén podemos definir unha función que dea como imaxe un vector de 2 ou máis valores. Por exemplo: $f := \mathbb{R}^2 \rightarrow \mathbb{R}^3$ definida por :

```
> f:=(x,y)->(x^2,x-1,exp(-x));
```

$$f := (x, y) \rightarrow (x^2, x - 1, e^{-x}) \tag{1.1.6}$$

```
> f(x,y);
```

$$x^2, x - 1, e^{-x} \tag{1.1.7}$$

```
> f(1,2);
```

$$1, 0, e^{-1} \tag{1.1.8}$$

Conversión de expresións en funcións

Para convertir unha expresión nunha función utilízase o comando:

unapply(expr, x, y, ..)

onde:

expr - expresión
x,y,.. - nomes de variabeis

```
> expresion:=(a^2*x^3+b*exp(t)+c^3*sin(x))/(a*x^2+c*t);
```

$$expresion := \frac{a^2 x^3 + b\,e^t + c^3 \sin(x)}{a\,x^2 + c\,t} \quad (1.2.1)$$

```
> f:=unapply(expresion, x, t); #crease unha función nas variabeis (x,t)
```

$$f := (x, t) \rightarrow \frac{a^2 x^3 + b\,e^t + c^3 \sin(x)}{a\,x^2 + c\,t} \quad (1.2.2)$$

```
> f(0,1);
```

$$\frac{b\,e}{c} \quad (1.2.3)$$

```
> restart; # limpa a memoria interna de MAPLE
```

Operacións sobre funcións

Con MAPLE pódese realizar as operacións de suma, multiplicación e composición de funcións da seguinte forma:

```
> f:=x->ln(x)+1; g:=y->exp(y)-1;
```

$$f := x \rightarrow \ln(x) + 1$$

$$g := y \rightarrow e^y - 1 \quad (1.3.1)$$

```
> h:=f+g; h(z);
```

$$h := f + g$$

$$\ln(z) + e^z \quad (1.3.2)$$

```
> h:=f*g; h(z);
```

$$h := f\,g$$

$$(\ln(z) + 1)\,(e^z - 1) \quad (1.3.3)$$

```
> h:=f@g; # composición de función co operador @
```

$$h := f@g \quad (1.3.4)$$

```
> h(z);
```

$$\ln(e^z - 1) + 1 \quad (1.3.5)$$

```
> h:=g@f; h(z);
```

$$h := g@f$$

$$e^{\ln(z) + 1} - 1 \quad (1.3.6)$$

```
> simplify(%); #simplifica a función anterior
```

$$z\,e - 1 \quad (1.3.7)$$

```
> (f@@4)(z); # equivalente a f(f(f(f(z))));
```

$$\ln(\ln(\ln(\ln(z) + 1) + 1) + 1) + 1 \quad (1.3.8)$$

Na composición temos que fixarnos que a segunda función poda aplicarse sobre a imaxen da primeira función aplicada. Por exemplo, que o nº de saídas da 1ª función (se é unha función vectorial) coincida co nº de entradas que espera a 2ª función).

Definición de funcións por intervalos

Hai que usar o comando:

piecewise(cond_1, f_1, cond_2, f_2, ..., cond_n, f_n, f_otherwise)

onde os parámetros son:

f_i - expresión

cond_i - intervalo de definición (relación ou combinación booleana de inecuacións)

f_otherwise - (opcional) expresión por defecto

```
> f:=x->piecewise(x<=1, x+1, 1<x and x<3, sin(x), 1);f(x);
```

$$f := x \rightarrow piecewise(x \le 1, x+1, 1 < x \text{ and } x < 3, \sin(x), 1)$$

$$\begin{cases} x+1 & x \le 1 \\ \sin(x) & 1 < x \text{ and } x < 3 \\ 1 & otherwise \end{cases} \tag{1.4.1}$$

```
> plot(f(x), x=-1..4);
```



Pódese facer exactamente igual cunha expresión:

```
> f:=piecewise(x<=1, x+1, 1<x and x<3, sin(x), 1);plot(f,x=-1.
.4);
```

$$f := \begin{cases} x+1 & x \le 1 \\ \sin(x) & 1 < x \text{ and } x < 3 \\ 1 & otherwise \end{cases}$$



Definición e representación gráfica de funcións de dúas variabeis

```
> f:=(x,y)->x^3-3*x*y^2;
```

(1.5.1)

$$f := (x, y) \rightarrow x^3 - 3\,x\,y^2 \tag{1.5.1}$$

```
> f(3,2); #evaluamos a función no punto (3,2)
```

$$-9 \tag{1.5.2}$$

Representamos a función de 2 variábeis co comando **plot3d**:

```
> plot3d(f,-1..1,-1..1, axes=FRAME, style=PATCH);
```



O mesmo podemos facer con expresións:

```
> f:=sin(x^2*y^2);plot3d(f,x=-Pi/2..Pi/2,y=-Pi/2..Pi/2);
```

$$f := \sin(x^2\,y^2)$$



Outro tipo de gráfica que permite representar en 2D funcións de dúas variábeis, é o **mapa de calor**, que mostra en vermello (resp. azul) as rexións do plano XY con valores elevados (resp. baixos) da función. Hai que usar o comando **contourplot**, no módulo **plots**:

```
> with(plots):contourplot(sin(x^2*y^2),x=-Pi/2..Pi/2,y=-Pi/2..
  Pi/2,filledregions=true);
```



Cálculo de límites

MAPLE pode calcular límites de expresións ou de funcións utilizando o comando

limit(f, x=a, dir) $\lim_{x \to a} f$

Limit(f, x=a, dir) (non evalúa o límite senon que so o deixa indicado)

onde os argumentos son:

f	-	expresión alxebráica
x	-	nome da variábel
a	-	punto do límite (pode ser infinity, ou -infinity)
dir	-	(opcional) refírese a dirección na que se calcula o límite, aproximando pola dereira (right) ou pola esquerda (left). Por defecto aproximase polos dous lados.

```
> limit( cos(x)/x , x=Pi/2);
```

$$0 \tag{1.6.1}$$

```
> limit( (-x^2+x+1)/(x+4) , x=infinity);
```

$$-\infty \tag{1.6.2}$$

```
> limit( tan(x) , x=Pi/2);
```

$$\mathit{undefined} \tag{1.6.3}$$

```
> limit( tan(x) , x=Pi/2 , right);
```

$$-\infty \tag{1.6.4}$$

```
> limit( tan(x) , x=Pi/2, left);
```

$$\infty \tag{1.6.5}$$

```
> Limit( cos(x)/x , x=Pi/2);
```

$$\lim_{x \to \frac{1}{2}\pi} \frac{\cos(x)}{x} \tag{1.6.6}$$

```
> value(%); # evalúase o límite anterior
```

$$0 \tag{1.6.7}$$

```
> Limit( tan(x) , x=Pi/2, left)=limit(tan(x),x=Pi/2,left);
```

$$\lim_{x \to \frac{1}{2}\pi^-} \tan(x) = \infty \tag{1.6.8}$$

```
> value(%);
```

$$\infty \tag{1.6.9}$$

Cálculo de derivadas

MAPLE pode calcula-la derivada dunha expresión con respecto a unha variábel dada utilizando o comando **Diff** ou **diff** ou o operador **D** (que opera sobre funcións). Comando **diff**:

diff(f, x1, ..., xj)	$\frac{d^j}{dx_j \ldots dx_1} f$
diff(f, [x1$n])	$\frac{d^n}{dx_1^n} f$
diff(f, x1$n, [x2$n, x3], ..., xj, [xk$m])	$\frac{d^r}{dx_k^m \, dx_j \ldots dx_3 \, dx_2^n \, dx_1^n} f$

onde:

f	-	expresión que se quere derivar
x1, x2, ..., xj	-	variabeis respecto as cales se calcula a derivada
n	-	orde de derivación

Tamén se pode utiliza-lo comando Diff co mesmo efecto que no caso do limite.

```
> f:=exp(-2*x);
```

$$f := e^{-2x} \tag{1.7.1}$$

```
> derivada:=diff(f,x); # devolve unha expresión
```

$$derivada := -2\, e^{-2x} \tag{1.7.2}$$

```
> f_prima:=unapply(derivada, x); # agora f_prima é unha
  función
```

$$f\_prima := x \rightarrow -2\, e^{-2x} \tag{1.7.3}$$

```
> f_prima(3); #evalúa a función nun punto
```

$$-2\, e^{-6} \tag{1.7.4}$$

```
> Diff(x^3+2*x , x);
```

$$\frac{d}{dx}\left(x^3 + 2\, x\right) \tag{1.7.5}$$

```
> value(%); # evalúa a expresión anterior
```

$$3\, x^2 + 2 \tag{1.7.6}$$

```
> Diff(x^3+2*x , x$2); #derivada segunda
```

$$\frac{d^2}{dx^2}\left(x^3 + 2\, x\right) \tag{1.7.7}$$

```
> value(%);
```

$$6\, x \tag{1.7.8}$$

O **diff** serve tamén para derivar funcións con respecto a varias variábeis ao mesmo tempo

```
> Diff(sin(x*y),x$2,y)=diff(sin(x*y),x$2,y);
```

$$\frac{\partial^3}{\partial y\, \partial x^2} \sin(x\, y) = -\cos(x\, y)\, x\, y^2 - 2 \sin(x\, y)\, y \tag{1.7.9}$$

Se queres calcular o valor da derivada nun punto, tes que empregar o **diff**

```
> subs(x=Pi,y=-Pi,diff(sin(x*y),x,y));
```

$$\sin(-\pi^2)\,\pi^2+\cos(-\pi^2) \quad (1.7.10)$$

Con funcións (definidas con ->) hai que empregar o operador **D**:

D(f) D[i](f) D[i](f)(x, y, ...)

onde os argumentos son:

f - función a derivar

i - enteiro ou enteiros, que indican a orde da variábel a respeito da cal se deriva (p.ex. se temos *f(x, y, z)*, entón **D[3,2,1](f)** é a derivada a respeito de *z*, *y*, *x*, sucesivamente

x, y, ... - punto de aplicación

O operador **D** é máis xeral que **diff** xa que pode evaluar a derivada nun punto.
Co comando **convert** pódese convertir un formato no outro.

```
> restart;
> D[1,1,2](g)(x,y);
```

$$D[1,1,2](g)(x,y) \quad (1.7.11)$$

```
> convert(%, diff);
```

$$\frac{\partial^3}{\partial y\,\partial x^2} g(x,y) \quad (1.7.12)$$

```
> Diff(x^3+2*x , x);
```

$$\frac{d}{dx}(x^3+2x) \quad (1.7.13)$$

```
> convert(%, D);
```

$$3x^2+2 \quad (1.7.14)$$

```
> restart;
> f:=x->ln(x)+sin(x); # definición dunha función
```

$$f := x \to \ln(x)+\sin(x) \quad (1.7.15)$$

```
> f_prima:=D(f); #devolve unha función
```

$$f\_prima := x \to \frac{1}{x}+\cos(x) \quad (1.7.16)$$

```
> g:=(x, y, z)->exp(x*y)+sin(x)*cos(z)+x*y*z; #función de varias variabeis
```

$$g := (x,y,z) \to e^{xy}+\sin(x)\cos(z)+x\,y\,z \quad (1.7.17)$$

```
> der:=D[2](g); #derivada respecto a y por ocupar-la segunda posición
```

$$der := (x,y,z) \to x\,e^{xy}+x\,z \quad (1.7.18)$$

Integración definida e indefinida

MAPLE realiza a integración definida e indefinida co comando **int**

int(expression,x)

	$\int expression\, dx$
int(expression,x=a..b)	$\int_a^b expression\, dx$
int(expression, [x = a..b, y = c..d, ...])	$\int_c^d \int_a^b expression\, dx dy$

onde os parámetros son:

expression	-	expresión a integrar
x, y	-	variabeis de integración
a, b, c, d	-	intervalo de integración no caso de integral indefinida

```
> restart;
> expresion:=2*x*exp(x^2);
```

$$expresion := 2\, x\, e^{x^2} \quad (1.8.1)$$

```
> int(expresion, x); # integral indefinida
```

$$e^{x^2} \quad (1.8.2)$$

```
> diff(%, x); # derivamos o resultado da integral e da
  expresion
```

$$2\, x\, e^{x^2} \quad (1.8.3)$$

```
> int(sin(y)*cos(y),y);
```

$$-\frac{1}{2}\cos(y)^2 \quad (1.8.4)$$

```
> int(1/exp(x^2)+x,x);
```

$$\frac{1}{2}\sqrt{\pi}\,\mathrm{erf}(x) + \frac{1}{2}x^2 \quad (1.8.5)$$

```
> int(1/x, x=2..4); # integral definida
```

$$\ln(2) \quad (1.8.6)$$

```
> expresion2:=1/(1+x^2);
```

$$expresion2 := \frac{1}{1+x^2} \quad (1.8.7)$$

```
> int(expresion2, x=0..infinity);
```

$$\frac{1}{2}\pi \quad (1.8.8)$$

O comando **Int** é interesante para visualizar a integral que estamos a realizar pero non avalía a expresión.

```
> Int(expresion2, x=0..infinity);
```

(1.8.9)

$$\int_{0}^{\infty} \frac{1}{1+x^2}\, dx \tag{1.8.9}$$

```
> value(%);
```

$$\frac{1}{2}\pi \tag{1.8.10}$$

Nota que a integral infinita é o límite da integral finita cando o intervalo de integración tende a infinito:

$$\int_{a}^{\infty} f(x)\, dx = \lim_{x \to \infty} \int_{a}^{x} f(t)\, dt$$

Por exemplo:

$$\int_{1}^{\infty} \frac{1}{x^2}\, dx = \lim_{x \to \infty} \int_{1}^{x} \frac{1}{t^2}\, dt$$

```
> int(1/t^2, t = 1..x)  assuming x :: positive
```

$$\frac{x-1}{x} \tag{1.8.11}$$

```
> limit(%, x = infinity);
```

$$1 \tag{1.8.12}$$

```
> int(1/x^2, x = 1..infinity);
```

$$1 \tag{1.8.13}$$

No caso de integrais dobres (é dicir, de funcións de dúas variábeis):

$$\int_{0}^{a}\int_{-x}^{x} (x^2 y + x y^2)\, dy\, dx = \int_{0}^{a}\left[\int_{-x}^{x} (x^2 y + x y^2)\, dy\right] dx$$

```
> Int(x^2*y+x*y^2, [y=-x..x,x=0..a])=int(int(x^2*y+x*y^2, y=-
  x..x),x=0..a);
```

$$\int_{0}^{a}\int_{-x}^{x} (x^2 y + x y^2)\, dy\, dx = \frac{2}{15} a^5 \tag{1.8.14}$$

```
> Int(x^2*y+x*y^2, [y=-x..x,x=0..a])=int(x^2*y+x*y^2, [y=-x..
  x,x=0..a]);
```

$$\int_{0}^{a}\int_{-x}^{x} (x^2 y + x y^2)\, dy\, dx = \frac{2}{15} a^5 \tag{1.8.15}$$

Dado que a integral debe ser un número que non pode depender de *x* nin de *y*, está claro que os límites variábeis (neste caso, - *x* e *x*) corresponden á variábel *y*, e os límites constantes (neste caso, *0* e *a*) corresponden a *x*. Sempre hai que integrar primeiro con respeito á variábel que ten os límites de integración variábeis (neste caso, *y).*